

```

> unprotect(Diff);
> Diff := proc(f::algebraic,x::name) local u,v,n,y;
  if type(f,numeric) then 0
  elif type(f,name) then if f=x then 1 else 0 fi
  elif op(0,f)='+' then add( Diff(u,x), u=f );
  elif op(0,f)='*' then
    u := op(1,f); v := subsop(1=1,f);
    Diff(u,x)*v + Diff(v,x)*u
  elif op(0,f)='^' and Diff(op(2,f),x)=0 then
    n := op(2,f); u := op(1,f);
    n*Diff(u,x)*u^(n-1)
  elif op(0,f)=exp then
    u := op(1,f); Diff(u,x)*exp(u);
  elif op(0,f)=int and type(op(2,f),name) then
    y := op(2,f); u := op(1,f);
    if x=y then u # Diff( int(u(x),x), x ) ==> u(x)
    else int( Diff(u,x), y ); # Diff( int(u(x,y),y), x )
    fi
  elif type(f,function) and nops(f)=1 and Diff(op(1,f),x)=0 then 0
  else 'Diff'(f,x)
  fi
end:

```

```

> Diff(2+x+3*x^2,x);

```

$$1 + 6x$$

```

> g := y*x^2*z-x*y+2*y*z;

```

$$g := yx^2z - xy + 2yz$$

```

> Diff(g,x);

```

$$2xyz - y$$

```

> Diff(u(x)/v(x)+1/w(x),x);

```

$$\frac{\frac{d}{dx} u(x)}{v(x)} - \frac{\left(\frac{d}{dx} v(x) \right) u(x)}{v(x)^2} - \frac{\frac{d}{dx} w(x)}{w(x)^2}$$

```

> Diff( exp(-2*x)+sin(2*x)+ln(2)*x, x );

```

$$-2e^{-2x} + \frac{d}{dx} \sin(2x) + \ln(2)$$

```

> g := int(2*x*f(t),t);

```

$$g := \int 2xf(t) dt$$

```

> Diff(g,t);

```

$$2xf(t)$$

```

> Diff(g,x);

```

$$\int 2f(t) dt$$